# $Id: PKGBUILD 183281 2013-04-20 16:41:14Z heftig $
# Maintainer: tobias <tobias@archlinux.org>
# Contributor: Dominik Ryba <domryba@post.pl>

pkgname=enblend-enfuse
pkgver=4.1.1
pkgrel=1
pkgdesc="Intelligent blend tool for overlapping picture"
arch=("i686" "x86_64")
license=('GPL')
url="http://enblend.sourceforge.net"
depends=('openexr' 'glew' 'freeglut' 'lcms2' 'boost-libs' 'vigra' 'gsl')
makedepends=('boost' 'mesa')
replaces=('enblend')
conflicts=('enblend')
provides=('enblend')
install=${pkgname}.install
source=(http://downloads.sourceforge.net/sourceforge/enblend/${pkgname}-${pkgver}.tar.gz)
sha1sums=('bc18fca3033b031d603b22678b2e680a0ffae1b6')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  ./configure --prefix=/usr
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make DESTDIR="${pkgdir}" install
  install -D -m644 doc/enblend.info "${pkgdir}/usr/share/info/enblend.info"
  install -D -m644 doc/enfuse.info "${pkgdir}/usr/share/info/enfuse.info"
}
